Homework #5 **4190.308 Computer Architecture**

Due Date: Wednesday, April 12, 2023 **Sample Solution**

Submission: electronically on eTL (scan & upload)

Problem 1

```
# long problem1(long a, long b, long c, long d)
# {
#   long p1 = 48*a + 36*b - c*d;
#   if (a << 1 > b >> 1) p1 -= c;
#   if ((b & 0xffff) > 0x8765) c += d;
#   p1 += c;
#
#   return p1;
# }


        .file   "p1.c"
        .text
        .align  2
        .globl  problem1
problem1:                       # a0 = a, a1 = b, a2 = c, a3 = d
        slli    a4,a0,1         # a4 = a << 1 = 2a
        add     a5,a4,a0        # a5 = 2a + a = 3a
        slli    a5,a5,4         # a5 = 3a << 4 = 48a
        slli    a0,a1,3         # a0 = b << 3 = 8b
        add     a0,a0,a1        # a0 = 8b + b = 9b
        slli    a0,a0,2         # a0 = 4b << 2 = 36b
        add     a5,a5,a0        # a5 = 48a + 36b
        mul     a0,a2,a3        # a0 = c*d
        sub     a5,a5,a0        # a5 = 48a + 36b - c*d =p1 (line 3)
        srai    a0,a1,1         # a0 = b >> 1 = b/2
        ble     a4,a0,.L2       # 2a <= b/2 ? goto .L2
        sub     a5,a5,a2        # a5 = p1 = p1 - c         (line 4)
.L2:
        slli    a1,a1,48        # a1 = b << 48
        srli    a1,a1,48        # a1 = (b << 48) >> 48 = b & 0xffff
        li      a4,32768        # a4 = 0x8000
        addi    a4,a4,1893      # a4 = 0x8000 + 0x765 = 0x8765
        ble     a1,a4,.L3       # (b & 0xffff) <= 0x8765 ? goto .L3
        add     a2,a2,a3        # a2 = c + d = c           (line 5)
.L3:
        add     a0,a2,a5        # p1 = a2(=c) + a3(=p1)
        ret                     # return p1
```

Problem 2

```
#  1 long problem2(int *a, long start, long stop,
#                  long exclude_start, long exclude_stop)
#  2 {
#  3   long chksum = -1;
#  4
#  5   if (!a) return chksum;
#  6
#  7   for (long p=start; p<stop; p++) {
#  8     if ((p < exclude_start) || (p >= exclude_stop)) {
#  9       chksum ^= a[p];
# 10     }
# 11   }
# 12
# 13   return chksum;
# 14 }

        .file   "p2.c"
        .text
        .align  2
        .globl  problem2
problem2:                       # a0 = *a, a1 = start, a2 = stop
                                # a3 = exclude_start, a4 = exclude_stop
        mv      a6,a0           # a6 = *a
        beq     a0,zero,.L6 # a == 0 ? goto .L6         (line 5)
        bge     a1,a2,.L7   # start >= stop ? goto .L7
        li      a0,-1       # a0 = -1                   (=chksum, line 3)
        j       .L5         # goto .L5

.L3:                            # update checksum (line 9)
        slli    a5,a1,2     # a5 = start << 2 = 4*start
        add     a5,a6,a5    # a5 = a + 4*start = &a[start]
        lw      a5,0(a5)    # load word (int) a5 = a[start]
        xor     a0,a0,a5    # chksum ^= a5              (line 9)

.L4:                            # for loop update and check (line 7)
        addi    a1,a1,1     # start++ = p++             (update)
        beq     a2,a1,.L9   # p == stop ? goto .L9      (cond. check)

.L5:                            # implementation of if (line 8)
        bgt     a3,a1,.L3   # exclude_start > p ? goto .L3 (1. cond check)
        bgt     a4,a1,.L4   # exclude_stop > p ? goto .L4  (2. check rev)
        j       .L3         # goto .L3                    (2. check true)
.L9:
        ret                 # return chksum in a0

.L6:                            # a==0: return -1           (line 3,5)
        li      a0,-1       # a0 = -1
        ret                 # return chksum in a0

.L7:                            # for -> do while check: return -1 (line 3,7)
        li      a0,-1       # a0 = -1
        ret                 # return chksum in a0
```